

Spice Models, Next Pages =>

```
****** Several TL431 Spice Models... I prefer the TI Version *****
********************************************************************
*                  REFERENCE
*                  |  ANODE
*                  |  |  CATHODE
*                  |  |  |
.SUBCKT  TL431/TI 1  2  3
V1  6  7  DC  1.4V
I1  2  4  1E-3
R1  1  2  1.2E6
R2  4  2  RMOD 2.495E3
R3  5  7  .2
D1  3  6  DMOD1
D2  2  3  DMOD1
D3  2  7  DMOD2
*E1  5  2  POLY(2)  (4,2)  (1,2)  0  710  -710
E1  5  2  VALUE = {V(4,2)*710 + V(1,2)*-710}
.MODEL RMOD RES (TC1=1.4E-5 TC2=-1E-6)
.MODEL DMOD1 D (RS=.3)
.MODEL DMOD2 D (RS=1E-6)
.ENDS
********************************************************************
.subckt TL431-R REFIN ANODE CATHODE
Q8 Q2_C Q2_C R4_N 0 P1
Q9 CATHODE Q7_C Q9_E 0 n1
R8 Q4_E ANODE 800
R9 Q2_E R9_N 4k
Q10 CATHODE R6_P ANODE 0 n1 5
Q11 Q7_C R10_P ANODE 0 n1
*.nodeset R3_N 1
C2 R9_N C2_N 20p
R10 R10_P R1_N 1k
C1 CATHODE Q7_C 20p
D1 ANODE Q7_C DIODE
D2 ANODE CATHODE DIODE
R4 CATHODE R4_N 800
Q2 Q2_C Q1_E Q2_E 0 n1
Q3 R9_N C2_N ANODE 0 n1
R5 CATHODE R5_N 800
R6 R6_P Q9_E 150
Q1 CATHODE REFIN Q1_E 0 n1
R7 ANODE R6_P 10k
Q6 Q7_C Q7_C REFIN 0 n1
Q7 Q7_C Q2_C R5_N 0 P1
R1 R3_N R1_N 2.4k
R2 R3_N C2_N 7.2k
Q4 C2_N R1_N Q4_E 0 n1
Q5 R1_N R1_N ANODE 0 n1
R3 Q1_E R3_N 3.28k
.model p1 pnp bf=50
.model n1 npn bf=100 tf=2n cjc=1p is=5e-18 NF=1.07
```

```
.model diode d rs=1 cjo=2p
.ends
********************************************************************
.SUBCKT TL431-X 7 6 11
*              K A FDBK
.MODEL DCLAMP D (IS=13.5N RS=25M N=1.59
+ CJO=45P VJ=.75 M=.302 TT=50.4N BV=34V IBV=1MA)
V1 1 6 2.495 ; used for fixed reference, replaced with EB1 Limiter
*EB1 1 6 Value = { IF ( V(7,6)> 2.495, 2.495, V(7,6) ) }
R1 6 2 15.6
C1 2 6 .5U
R2 2 3 100
C2 3 4 .08U
R3 4 6 10
G2 6 8 3 6 1.73
D1 5 8 DCLAMP
D2 7 8 DCLAMP
V4 5 6 2
G1 6 2 1 11 0.11
.ENDS
********************************************************************
.SUBCKT AEITL431AILP 7 6 11
*Simplified TL431 model
.MODEL DCLAMP D (IS=13.5N RS=25M N=1.59
+ CJO=45P VJ=.75 M=.302 TT=50.4N BV=34V IBV=1MA)
V1 1 6 2.495
R1 6 2 15.6
C1 2 6 .5U
R2 2 3 100
C2 3 4 .08U
R3 4 6 10
G2 6 8 3 6 1.73
D1 5 8 DCLAMP
D2 7 8 DCLAMP
V4 5 6 2
G1 6 2 1 11 0.11
.ENDS
********************************************************************
```